@echo off
setlocal

rem Nombre de los instaladores (editar de ser necesario)

set JDK_INSTALLER=jdk-6u26-windows-i586.exe
set IDEA_INSTALLER=ideaIC-10.5.1.exe
set GROOVY_ZIP=groovy-binary-1.8.1.zip
set UNZIP=unzip.exe

rem Ubicaciones destino para los programas

set JAVA_DIR=c:\java
set JDK_DIR=%JAVA_DIR%\jdk\1.6.0
set JRE_DIR=%JAVA_DIR%\jre\1.6.0
set IDEA_DIR=%JAVA_DIR%\idea\10ce

rem -----------------------------------------------------------


if not exist %JDK_INSTALLER% goto noinstallers
if not exist %IDEA_INSTALLER% goto noinstallers
if not exist %GROOVY_ZIP% goto noinstallers

echo Instalando JDK 1.6.0 en %JDK_DIR%
%JDK_INSTALLER% /s /INSTALLDIRPUBJRE="%JRE_DIR%" INSTALLDIR="%JDK_DIR%"

echo Instalando IntelliJ Idea 10 en %IDEA_DIR%
%IDEA_INSTALLER% /S /D=%IDEA_DIR%

echo Descomprimiendo los archivos de Groovy
set CURDIR="%CD%"
cd\
%JDK_DIR%\bin\jar xvf %CURDIR%\%GROOVY_ZIP% 
cd %CURDIR%

echo Instalacion finalizada
goto end

:noinstallers
echo Ejecute este .BAT en el mismo directorio donde se encuentran los instaladores
echo.
echo Los siguientes programas y archivos de instalacion son requeridos:
echo.
echo %JDK_INSTALLER%
echo %IDEA_INSTALLER%
echo %GROOVY_ZIP%
echo.
echo Para instrucciones de donde bajar estos programas ver:
echo http://goo.gl/zmSDA
echo.
echo Si el nombre de alguno de los instaladores que bajo de internet
echo difiere, puede cambiarlo editando este .BAT

:end
endlocal
echo.